#!/bin/zsh

### Get the path to a plugin's init file.
#function __antidote_initfiles {
  emulate -L zsh; setopt local_options $_adote_funcopts
  typeset -ga reply=()
  local dir=${1:A}
  local initfiles=($dir/${dir:A:t}.plugin.zsh(N))
  [[ $#initfiles -gt 0 ]] || initfiles=($dir/*.plugin.zsh(N))
  [[ $#initfiles -gt 0 ]] || initfiles=($dir/*.zsh(N))
  [[ $#initfiles -gt 0 ]] || initfiles=($dir/*.sh(N))
  [[ $#initfiles -gt 0 ]] || initfiles=($dir/*.zsh-theme(N))

  typeset -ga reply=($initfiles)
  printf "%s\n" ${(u)initfiles[@]}
  (( $#initfiles )) || return 1
#}
